USING SYMBOLIC COMPUTER ALGEBRAIC SYSTEMS TO DERIVE FORMULAS INVOLVING ORTHOGONAL POLYNOMIALS AND OTHER SPECIAL FUNCTIONS

GEORGE GASPER*
Department of Mathematics
Northwestern University
Evanston, IL 60208

d in Orthogonal Polynomials: Theory and Practice, ed. by P. Nevai, Kluwer Academic Publishers, Boston (1990), pp. 163 - 179

ABSTRACT. It is shown how symbolic computer algebraic systems such as Mathematica, Macsyma, SMP, etc., can be used to derive transformation and expansion formulas for orthogonal polynomials that are expressible in terms of either hypergeometric or basic hypergeometric series. In particular, we demonstrate how Mathematica can be used to apply transformation formulas to the Racah and q-Racah polynomials, to derive an indefinite bibasic summation formula, an expansion formula for Laguerre polynomials, Clausen's formula for the square of hypergeometric series, a q-analogue of a Fields and Wimp expansion formula, and to prove the Askey-Gasper inequality which de Branges used in his proof of the Bieberbach conjecture. We also make some observations and conjectures related to Jensen's necessary and sufficient conditions for the Riemann Hypothesis to hold.

1. Introduction

Now that several symbolic computer algebraic systems such as Derive, Reduce, Scratchpad, SMP, and the three M's "Macsyma, Maple, and Mathematica" are available for various computers, it is natural for persons having access to such a system to try to have it perform the tedious symbolic manipulations needed to derive certain formulas involving orthogonal polynomials and other special functions. Here, for definiteness, we will use Mathematica to illustrate how the author has been employing it (and Macsyma, SMP, etc.) to derive transformation and expansion formulas for orthogonal polynomials that are expressible in terms of either hypergeometric series

$$(1.1) \qquad {}_rF_s\left[\begin{matrix} a_1, \ldots, a_r \\ b_1, \ldots, b_s \end{matrix}; z\right] = \sum_{n=0}^{\infty} \frac{(a_1)_n \cdots (a_r)_n}{n!(b_1)_n \cdots (b_s)_n}\, z^n$$

*This work was supported in part by a grant from the National Science Foundation under grant DMS-8601901.

or q-(basic) hypergeometric series

$$
(1.2) \qquad {}_r\phi_s \left[\begin{matrix} a_1, \dots, a_r \\ b_1, \dots, b_s \end{matrix} ; q, z \right] = \sum_{n=0}^{\infty} \frac{(a_1;q)_n \cdots (a_r;q)_n}{(q;q)_n (b_1;q)_n \cdots (b_s;q)_n} \left[(-1)^n q^{n(n-1)/2}\right]^{1+s-r} z^n
$$

where $(a)_n = \prod_{k=0}^{n-1}(a+k)$ is the shifted factorial (Pochhammer symbol) and $(a;q)_n = \prod_{k=0}^{n-1}(1-aq^k)$ is the q-shifted factorial. Unless stated otherwise we will assume that $0 < |q| < 1$. For such orthogonal polynomials, which include the classical orthogonal polynomials of Gegenbauer, Hahn, Hermite, Jacobi, Krawtchouk, Laguerre, Meixner, Tchebichef, and their q-analogues, see Askey and Wilson [5], Chihara [10], Erdélyi [14], Gasper and Rahman [27], and Szegö [30].

Although our methods are applicable to all of the above mentioned orthogonal polynomials, in order to demonstrate them in this paper we will only consider certain formulas involving the Laguerre polynomials

$$
(1.3) \qquad L_n^a(x) = \frac{(a+1)_n}{n!} \, {}_1F_1 \left[\begin{matrix} -n \\ a+1 \end{matrix} ; x \right]
$$

the Gegenbauer (ultraspherical) polynomials

$$
(1.4) \qquad C_n^a(x) = \frac{(2a)_n}{n!} \, {}_2F_1 \left[\begin{matrix} -n, n+2a \\ a+1/2 \end{matrix} ; \frac{1-x}{2} \right]
$$

the Racah polynomials

$$
(1.5) \qquad W_n(x;a,b,c,N) = {}_4F_3 \left[\begin{matrix} -n, n+a+b+1, -x, c+x-N \\ a+1, b+c+1, -N \end{matrix} ; 1 \right]
$$

and the q-Racah polynomials

$$
(1.6) \qquad W_n(x;a,b,c,N;q) = {}_4\phi_3 \left[\begin{matrix} q^{-n}, abq^{n+1}, q^{-x}, cq^{x-N} \\ aq, bcq, q^{-N} \end{matrix} ; q, q \right].
$$

In §2 we illustrate how the symbolic manipulation capabilities of Mathematica can be utilized to apply certain transformation formulas to the Racah and q-Racah polynomials. Symbolic factorization is employed in §3 to derive an indefinite bibasic summation formula. A technique for manipulating multiple series is demonstrated in §4 by using it to derive the expansion formula

$$
(1.7) \qquad L_n^b(x) = \sum_{k=0}^{n} \frac{(b-a)_{n-k}}{(n-k)!} L_k^a(x).
$$

The coefficients in (1.7) are called the connection coefficients between the sequences $\{L_n^b(x)\}$ and $\{L_n^a(x)\}$. Notice that the above connection coefficients are clearly nonnegative when $a \le b$. For applications of the nonnegativity of the connection coefficients between two sequences of orthogonal polynomials to positive definite functions, isometric embeddings of metric spaces, the derivation of inequalities, etc., see Askey [1], Askey and Gasper [2], Gangolli [16], Gasper [17], [18], and Gasper and Rahman [27].

Multiple series manipulations are also used in §5 to derive Clausen's [11] formula

$$\left\{ {}_2F_1\left[\begin{matrix} a, b \\ a+b+1/2 \end{matrix}; x\right]\right\}^2 = {}_3F_2\left[\begin{matrix} 2a, 2b, a+b \\ 2a+2b, a+b+1/2 \end{matrix}; x\right] \tag{1.8}$$

and, in §6, to derive a q-extension (q-analogue) of the Fields and Wimp [15] expansion formula

$$\begin{aligned} {}_{r+t}F_{s+u}\left[\begin{matrix} a_R, c_T \\ b_S, d_U \end{matrix}; xw\right] &= \sum_{n=0}^{\infty} \frac{(c_T)_n (e_K)_n (-x)^n}{(d_U)_n (f_M)_n (n+\gamma)_n n!} \\ &\cdot {}_{k+t}F_{m+u+1}\left[\begin{matrix} n+c_T, n+e_K \\ 2n+1+\gamma, n+d_U, n+f_M \end{matrix}; x\right] \\ &\cdot {}_{m+r+2}F_{k+s}\left[\begin{matrix} -n, n+\gamma, a_R, f_M \\ b_S, e_K \end{matrix}; w\right] \end{aligned} \tag{1.9}$$

where we employed the contracted notation of representing $a_1, a_2, \ldots, a_r$ by a_R, $(a_1)_n(a_2)_n \cdots (a_r)_n$ by $(a_R)_n$, and $n+a_1, n+a_2, \ldots, n+a_r$ by $n+a_R$. In (1.9), as elsewhere, either the parameters and variables are assumed to be such that the (multiple) series converge absolutely or the series are considered to be formal power series in the variables x and w. As an application of (1.8) and (1.9) we point out how they can be used to prove the Askey-Gasper [3], [4] inequality

$${}_3F_2\left[\begin{matrix} -n, n+a+2, (a+1)/2 \\ a+1, (a+3)/2 \end{matrix}; x\right] \geq 0, \qquad 0 \leq x \leq 1, \tag{1.10}$$

for $a > -2$ and $n = 0, 1, \ldots$, which was used by de Branges [7], [8] in his proof of the Bieberbach conjecture.

In §7 we make some observations and conjectures related to Jensen's necessary and sufficient conditions for the Riemann Hypothesis to hold.

2. Transformation formulas

In order to use Mathematica to apply Whipple's ${}_4F_3$ transformation formula [6, 7.2(1)]

$$\begin{aligned} &{}_4F_3\left[\begin{matrix} -n, a, b, c \\ d, e, f \end{matrix}; 1\right] \\ &= \frac{(e-a)_n (f-a)_n}{(e)_n (f)_n}\, {}_4F_3\left[\begin{matrix} -n, a, d-b, d-c \\ d, a+1-n-e, a+1-n-f \end{matrix}; 1\right] \end{aligned} \tag{2.1}$$

where $d+e+f = a+b+c+1-n$, to the Racah polynomials we let $p[a, n]$ symbolically denote the Pochhammer symbol $(a)_n$ (without defining it as a product) and enter the following function definition into a Mathematica session

In[1] := **transform4F3[mn_,a_,b_,c_,d_,e_,f_] := p[e-a,-mn] p[f-a,-mn]*
fourF3[mn,a,d-b,d-c,d,a+1+mn-e,a+1+mn-f]/(p[e,-mn] p[f,-mn])**

which represents the right side of (2.1). The a_, b_, etc., on the left side refer to any expressions, to be named a, b, etc., and the := defines the transformation rule to be used automatically each time the left side is requested. Either a * or a space may be used between variables and functions to denote multiplication in Mathematica, and, as above, the * has to be used when the next factor is continued on the next line. Notice that we used the symbol mn to denote the $-n$ argument on the left side of (2.1). Since Mathematica's built-in functions begin with capital letters (function names cannot begin with numbers and all function arguments must be enclosed in square brackets), we chose function names that begin with lower case letters to prevent any possible confusion with Mathematica's built-in functions.

In view of the ${}_4F_3$ series representation for the Racah polynomials in (1.5) we enter

In[2] := **transform4F3[-n,n+a+b+1,-x,c+x-N,a+1,b+c+1,-N]**

and then Mathematica responds with

Out[2]= (fourF3[$-n, 1+a+b+n, 1+a+x, 1+N+a-c-x, 1+a, 1+a-c,$
$2+N+a+b$] $p[-a+c-n,\ n]$ $p[-1-N-a-b-n,\ n]$)/
($p[-N,\ n]$ $p[1+b+c,\ n]$)

which is the ${}_4F_3$ series representation for the Racah polynomials that results when Whipple's transformation formula (2.1) is applied to the ${}_4F_3$ on the right side of (1.5).

Analogously, to apply Sears' [29] q-analogue of Whipple's formula

(2.2)
$$
\begin{aligned}
&{}_4\phi_3\left[\begin{matrix} q^{-n}, a, b, c \\ d, e, f \end{matrix}; q, q\right] \\
&= \frac{(e/a;q)_n (f/a;q)_n}{(e;q)_n (f;q)_n} a^n {}_4\phi_3\left[\begin{matrix} q^{-n}, a, d/b, d/c \\ d, aq^{1-n}/e, aq^{1-n}/f \end{matrix}; q, q\right]
\end{aligned}
$$

where $def = abcq^{1-n}$, to the q-Racah polynomials we let $o[a,q,n]$ symbolically denote the q-shifted factorial $(a;q)_n$ and enter the function definition

In[3] := **transform4phi3[mn_,a_,b_,c_,d_,e_,f_,q_] := o[e/a,q,-mn] o[f/a,q,-mn]*
a^(-mn) fourphi3[q^mn,a,d/b,d/c,d,a q^(1+mn)/e,a q^(1+mn) /f,q,q]/
(o[e,q,-mn] o[f,q,-mn])**

Then, entering

In[4] := **transform4phi3[-n,a b q^(n+1),q^(-x),c q^(x-N),a q,b c q,q^(-N),q]**

yields the ${}_4\phi_3$ series representation

Out[4] = ($a^n\ b^n\ q^{n\ (1\ +\ n)}$ fourphi3[$q^{-n}, a\ b\ q^{1\ +\ n},\ a\ q^{1\ +\ x},\ \dfrac{a\ q^{1\ +\ N\ -\ x}}{c}, a\ q,\ \dfrac{a\ q}{c},$
$a\ b\ q^{2\ +\ N},\ q,\ q$] $o[\dfrac{c}{a\ q^n},\ q,\ n]$ $o[\dfrac{q^{-1\ -\ N\ -\ n}}{a\ b},\ q,\ n]$)/
($o[b\ c\ q,\ q,\ n]$ $o[q^{-N},\ q,\ n]$)

for the q-Racah polynomials.

Several other formulas for the Racah and q-Racah polynomials may be obtained by applying these transformations to the corresponding series on the right sides of (1.5) and (1.6) with their second, third and fourth numerator arguments interchanged or their denominator arguments interchanged, and by iterating these transformations. On a Macintosh II with five megabytes of RAM, it only took Mathematica about one-half second and one second, respectively, of CPU time to compute Out[2] and Out[4]. Application of Watson's transformation formula [30, (3.4.1.5)]

$$
\begin{aligned}
{}_4\phi_3\left[\begin{matrix} q^{-n}, a, b, c \\ d, e, f \end{matrix}; q, q\right] &= \frac{(d/b;q)_n (d/c;q)_n}{(d;q)_n (d/bc;q)_n} \\
&\cdot {}_8\phi_7\left[\begin{matrix} \sigma, q\sqrt{\sigma}, -q\sqrt{\sigma}, f/a, e/a, b, c, q^{-n} \\ \sqrt{\sigma}, -\sqrt{\sigma}, e, f, ef/ab, ef/ac, efq^n/a \end{matrix}; q, \frac{efq^n}{bc}\right]
\end{aligned}
\tag{2.3}
$$

where $def = abcq^{1-n}$ and $\sigma = ef/aq$, to the q-Racah polynomials took about two seconds. One advantage of using a symbolic computer algebraic system is that once a function is defined its definition can be stored in a file (called a Notebook in Mathematica) from which it can be quickly read into any other session whenever it is needed. Another advantage is that one can also have the computer automatically check that the required "balanced" conditions $d + e + f = a + b + c + 1 - n$ in (2.1) and $def = abcq^{1-n}$ in (2.2) or (2.3) are satisfied before it computes the transformations. For example, rather than using the definition given in In[3] it is preferable to replace it by

```
transform4phi3[mn_,a_,b_,c_,d_,e_,f_,q_] :=
      If [d e f == a b c q^(1+mn), o[e/a,q,-mn] o[f/a,q,-mn] a^(-mn)*
      fourphi3[q^mn,a,d/b,d/c,d,a q^(1+mn)/e,a q^(1+mn)/ f,q,q]/
      (o[e,q,-mn] o[f,q,-mn]),
      Print["ERROR — Nonbalanced Series"],
      Print["ERROR — Nonbalanced Series"]]
```

which will immediately print "ERROR — Nonbalanced Series" whenever the required "balanced" condition is not satisfied. In the above display the logical operator == tests whether the expressions on the left and right sides of it are equal. For a limited time persons who wish to obtain copies of my Mathematica input files (which can easily be converted to work with Macsyma) containing symbolic forms of most of the identities and the summation and transformation formulas in the three Appendices of the book [27] may obtain them via email by contacting me at either gasper@ nuacc.bitnet or george@ math.nwu.edu, or by mailing me a formatted Macintosh 3.5" disk along with a self-addressed stamped envelope.

3. An indefinite bibasic summation formula

As in Gasper [21], let

$$
s_k = \frac{(ap;p)_k (bp;p)_k (cq;q)_k (aq/bc;q)_k}{(q;q)_k (aq/b;q)_k (ap/c;p)_k (bcp;p)_k}
\tag{3.1}
$$

for $k = 0, 1, 2, \ldots, s_{-1} = 0$, and define the difference operator Δ by $\Delta s_k = s_k - s_{k-1}$. If we enter the Product definition for s_k into Mathematica and apply the Factor command, we only get the difference of the two products that we started with. Therefore, we first observe that

$$\Delta s_k = \frac{(ap;p)_{k-1}(bp;p)_{k-1}(cq;q)_{k-1}(aq/bc;q)_{k-1}}{(q;q)_k(aq/b;q)_k(ap/c;p)_k(bcp;p)_k} \cdot \big[(1-ap^k)(1-bp^k)(1-cq^k)(1-aq^k/bc) - (1-q^k)(1-aq^k/b)(1-ap^k/c)(1-bcp^k)\big] \tag{3.2}$$

and then ask Mathematica to factor the above term in square brackets by entering

In[1] := **Factor[(1-a p^k)(1-b p^k)(1-c q^k)(1-a q^k/b/c)-**
(1-q^k)(1-a q^k/b)(1-a p^k/c)(1-b c p^k)]

to obtain

$$\textit{Out[1]} = \frac{(-1+c)\ (1-a\ p^k\ q^k)\ (-a+b\ c)\ (b\ p^k-q^k)}{b\ c}$$

which combined with (3.2) shows that

$$\Delta s_k = \frac{(1-ap^kq^k)(1-bp^kq^{-k})}{(1-a)(1-b)} \frac{(a;p)_k(b;p)_k(c;q)_k(a/bc;q)_k}{(q;q)_k(aq/b;q)_k(ap/c;p)_k(bcp;p)_k} q^k. \tag{3.3}$$

Since $\Delta s_0 = s_0 = 1$ and

$$\sum_{k=0}^{n} \Delta s_k = s_n \tag{3.4}$$

for $n \geq 0$, it follows from (3.3) that we have the indefinite bibasic summation formula [21]

$$\sum_{k=0}^{n} \frac{(1-ap^kq^k)(1-bp^kq^{-k})}{(1-a)(1-b)} \frac{(a;p)_k(b;p)_k(c;q)_k(a/bc;q)_k}{(q;q)_k(aq/b;q)_k(ap/c;p)_k(bcp;p)_k} q^k = \frac{(ap;p)_n(bp;p)_n(cq;q)_n(aq/bc;q)_n}{(q;q)_n(aq/b;q)_n(ap/c;p)_n(bcp;p)_n}. \tag{3.5}$$

Observing that $(q^{1-n};q)_n = 0$ for $n \geq 1$, we find that when $c = q^{-n}$, $n = 0, 1, 2, \ldots$, formula (3.5) reduces to the summation formula

$$\sum_{k=0}^{n} \frac{(1-ap^kq^k)(1-bp^kq^{-k})}{(1-a)(1-b)} \frac{(a;p)_k(b;p)_k(q^{-n};q)_k(aq^n/b;q)_k}{(q;q)_k(aq/b;q)_k(apq^n;p)_k(bpq^{-n};p)_k} q^k = \delta_{n,0} \tag{3.6}$$

for $n = 0, 1, 2, \ldots$, where $\delta_{n,m}$ is the Kronecker delta function. Formula (3.6) was independently derived in an equivalent form by Bressoud [9, §4]. This formula will be employed in our derivation in §6 of the q-analogue of the Fields and Wimp formula (1.9). The above derivation of (3.5) can be extended to give the generalization derived in Gasper and Rahman [25, (1.7)].

4. Derivation of the Laguerre polynomial expansion formula (1.7)

Although Mathematica's Sum[c[i, j, ...], {i, imin, imax}, {j, jmin, jmax}, ...] can be used to represent a multiple series such as those on the right sides of (1.7) and (1.9), one runs into difficulties in trying to perform the series manipulations needed to derive formulas involving multiple series such as those in [18], [19], [20], [21]. Here we will use the expansion formula (1.7) to demonstrate how one can derive formulas involving multiple series by working directly with the terms in the series.

Since the use of an identity such as

$$(a)_{n+k} = (a)_n(a+n)_k \tag{4.1}$$

to replace $(a)_{n+k}$ by $(a)_n(a+n)_k$ corresponds to multiplying by 1 in the form

$$(a)_n(a+n)_k/(a)_{n+k} \tag{4.2}$$

and cancelling the two $(a)_{n+k}$ products, we start by entering the following defined function into Mathematica

In[1] := **asnpk[a_,n_,k_] := p[a,n] p[a+n,k]/p[a,n+k]**

Then asnpk[a, n, k] equals 1 for all choices of its arguments, and multiplication of an expression such as a term in a series by this function followed by symbolic cancellation corresponds to using the identity (4.1). Sometimes the Simplify command has to be applied so that the desired symbolic cancellations occur and the simplest expressions are formed. The function name asnpk is a mnemonic for "a sub n plus k," which makes it easy to remember that multiplication by this function corresponds to applying the identity (4.1). Similarly, to apply the identity

$$(-n)_k = (-1)^k(1)_n/(1)_{n-k} \tag{4.3}$$

we enter the definition

In[2] := **mnsk[mn_,k_] := (-1)^k p[1,-mn]/(p[1,-mn-k] p[mn,k])**

where mnsk stands for "minus n sub k." Then mnsk[-n, k] equals 1 for $k = 0, 1, \dots, n$. Notice that even if we multiply an expression by one of these functions after making a mistake in typing the desired choice of arguments, we will still get a correct answer (i.e., an equal expression) because we only multiplied by 1.

Analogously, to apply a summation formula

$$\sum_k a_k = A \tag{4.4}$$

with $A \neq 0$, we first rewrite it in the form

$$\sum_k A^{-1}a_k = 1 \tag{4.5}$$

and then represent this series by its k^{th} term $A^{-1}a_k$. For example, with Vandermonde's summation formula [6, p. 3]

$$(4.6) \qquad {}_2F_1\left[\begin{matrix}-n, a\\ c\end{matrix}; 1\right] = \frac{(c-a)_n}{(c)_n}$$

rewritten in the form

$$(4.7) \qquad \sum_{k=0}^{n} \frac{(c)_n(-n)_k(a)_k}{(c-a)_n(1)_k(c)_k} = 1$$

we define

In[3] := **vand2F1[mn_,a_,c_,k_] := p[c,-mn] p[mn,k] p[a,k]/
(p[c-a,-mn] p[1,k] p[c,k])**

so that the sum of vand2F1[-n,a,c,k] from $k = 0$ to n equals 1 for all values of a and c when $n = 0, 1, 2, \ldots$. Similarly, to be able to apply the expansion

$$(4.8) \qquad x^j = \sum_{k=0}^{j} \frac{(a+1)_j(-j)_k}{(a+1)_k} L_k^a(x)$$

we define the function

In[4] := **jthpowerofx[j_,x_,a_,k_] := x^(-j) p[a+1,j] p[-j,k] laguerre[k,a,x] /
p[a+1,k]**

whose sum over all k equals 1.

Now we are ready to derive formula (1.7). Enter

In[5] := **p[b+1,n] p[-n,j] x^j/(p[1,n] p[1,j] p[b+1,j])**

to get

$$Out[5] = \frac{x^j\ p[-n,\ j]\ p[1+b,\ n]}{p[1,\ j]\ p[1,\ n]\ p[1+b,\ j]}$$

which is the j^{th} term of the series representation for the Laguerre polynomial $L_n^b(x)$ on the left side of (1.7). Then

In[6] := **% jthpowerofx[j,x,a,k]**

where % stands for the last result (Out[5] in this case which could have been used instead of the %), gives

$$Out[6] = \frac{\text{laguerre}[k,\ a,\ x]\ p[-j,\ k]\ p[-n,\ j]\ p[1+a,\ j]\ p[1+b,\ n]}{p[1,\ j]\ p[1,\ n]\ p[1+a,\ k]\ p[1+b,\ j]}$$

which is the j, k^{th} term in the double sum obtained by using the expansion (4.8) in the series for $L_n^b(x)$. Apply the identity (4.3) by using

In[7] := **% mnsk[-j,k]**

to obtain

$$Out[7] = \frac{(-1)^k \text{ laguerre}[k,\ a,\ x]\ p[-n,\ j]\ p[1+a,\ j]\ p[1+b,\ n]}{p[1,\ n]\ p[1,\ j-k]\ p[1+a,\ k]\ p[1+b,\ j]}$$

and then use

In[8] := **%/. j->j+k**

to replace j by $j+k$ and get

$$Out[8] = \frac{(-1)^k \text{ laguerre}[k,\ a,\ x]\ p[-n,\ j+k]\ p[1+a,\ j+k]\ p[1+b,\ n]}{p[1,\ j]\ p[1,\ n]\ p[1+a,\ k]\ p[1+b,\ j+k]}$$

In view of the above $j+k$'s we can use

In[9] := **% asnpk[-n,k,j] asnpk[a+1,k,j]/asnpk[b+1,k,j]**

to obtain

$$Out[9] = \frac{(-1)^k \text{ laguerre}[k,\ a,\ x]\ p[-n,\ k]\ p[1+b,\ n]\ p[k-n,\ j]\ p[1+a+k,\ j]}{p[1,\ j]\ p[1,\ n]\ p[1+b,\ k]\ p[1+b+k,\ j]}$$

Noticing that Vandermonde's summation formula may be applied to evaluate the sum over j, we enter

In[10] := **%/vand2F1[k-n,1+a+k,1+b+k,j]**

to get

$$Out[10] = \frac{(-1)^k \text{ laguerre}[k,\ a,\ x]\ p[-n,\ k]\ p[1+b,\ n]\ p[-a+b,\ -(k-n)]}{p[1,\ n]\ p[1+b,\ k]\ p[1+b+k,\ -(k-n)]}$$

Finally, by entering

In[11] := **Simplify[% asnpk[b+1,k,n-k] mnsk[-n,k] (-1)^(-2k)]**

we obtain

$$Out[11] = \frac{\text{laguerre}[k,\ a,\ x]\ p[-a+b,\ -k+n]}{p[1,\ -k+n]}$$

whose sum over k is the right side of (1.7), which concludes our derivation of (1.7).

5. Derivation of Clausen's formula

Observing that

$$\left\{ {}_2F_1\left[\begin{matrix} a, b \\ a+b+1/2 \end{matrix}; x\right]\right\}^2 = \sum_{j=0}^{\infty}\sum_{k=0}^{\infty} \frac{(a)_j (b)_j (a)_k (b)_k x^{j+k}}{(1)_j (a+b+1/2)_j (1)_k (a+b+1/2)_k} \tag{5.1}$$

we start by entering

In[1] := **p[a,j] p[b,j] p[a,k] p[b,k] x^(j+k)/**
(p[1,j] p[a+b+1/2,j] p[1,k] p[a+b+1/2,k])

and obtaining

$$Out[1] = \frac{x^{j\,+\,k}\ p[a,\ j]\ p[a,\ k]\ p[b,\ j]\ p[b,\ k]}{p[1,\ j]\ p[1,\ k]\ p\ [\frac{1}{2}+a+b,\ j]\ p[\frac{1}{2}+a+b,\ k]}$$

which is the j, k^{th} term of the double series on the right side of (5.1). In view of the $j+k^{\text{th}}$ power of x, we apply the substitution

In[2] : **%/. j->j-k**

to obtain

$$Out[2] = \frac{x^{j}\ p[a,\ k]\ p[a,\ j-k]\ p[b,\ k]\ p[b,\ j-k]}{p[1,\ k]\ p[1,\ j-k]\ p[\frac{1}{2}+a+b,\ k]\ p[\frac{1}{2}+a+b,\ j-k]}$$

To be able to apply the identity

$$(a)_{n-k} = (-1)^k (a)_n / (1-n-a)_k \tag{5.2}$$

enter the definition

In[3] := **asnmk[a_,n_,k_] := (-1)^k p[a,n]/(p[1-n-a,k] p[a,n-k])**

where asnmk stands for a sub n minus k. Then

In[4] := **Out[2] asnmk[a,j,k] asnmk[b,j,k]/**
(asnmk[1,j,k] asnmk[a+b+1/2,j,k])

gives

$$Out[4] = \frac{x^{j}\ p[a,\ j]\ p[a,\ k]\ p[b,\ j]\ p[b,\ k]\ p[-j,\ k]\ p[\frac{1}{2}-a-b-j,\ k]}{p[1,\ j]\ p[1,\ k]\ p[\frac{1}{2}+a+b,\ j]\ p[\frac{1}{2}+a+b,\ k]\ p[1-a-j,\ k]\ p[1-b-j,\ k]}$$

To sum over k it suffices to apply the summation formula

$${}_4F_3\left[\begin{matrix}-n, a, b, 1/2-a-b-n\\ a+b+1/2, 1-a-n, 1-b-n\end{matrix}; 1\right] = \frac{(2a)_n (2b)_n (a+b)_n}{(a)_n (b)_n (2a+2b)_n} \tag{5.3}$$

which follows easily from Dougall's ${}_7F_6$ summation formula [6, 4.3.(5)]. Enter this formula into Mathematica by defining

In[5] := **specialsum[n_,a_,b_,k_] := p[-n,k] p[a,k] p[b,k] p[1/2-a-b-n,k]***
p[a,n] p[b,n] p[2a+2b,n]/(p[1,k] p[a+b+1/2,k] p[1-a-n,k]*
p[1-b-n,k] p[2a,n] p[2b,n] p[a+b,n])

whose sum over k equals 1. Then,

In[6] := **Out[4]/specialsum[j,a,b,k]**

yields

$$Out[6] = \frac{x^j\ p[2\ a,\ j]\ p[2\ b,\ j]\ p[a+b,\ j]}{p\ [1,\ j]\ p[2\ a+2\ b,\ j]\ p[\frac{1}{2}+a+b,\ j]}$$

which is the j^{th} term of the ${}_3F_2$ series on the right side of (1.8). This completes our derivation of Clausen's formula.

One particularly important special case of Clausen's formula is that for the ultraspherical polynomials it gives the formula

$$\{C_n^a(x)\}^2 = \left(\frac{(2a)_n}{n!}\right)^2 {}_3F_2\left[\begin{matrix}-n, n+2a, a\\ 2a, a+1/2\end{matrix}; 1-x^2\right] \tag{5.4}$$

by using the series representation

$$C_n^a(x) = \frac{(2a)_n}{n!}\, {}_2F_1\left[\begin{matrix}-n/2, a+n/2\\ a+1/2\end{matrix}; 1-x^2\right]. \tag{5.5}$$

An extension of (5.4) to the continuous q-ultraspherical polynomials

$$C_n(\cos\theta;\beta|q) = \frac{(\beta^2;q)_n}{\beta^{n/2}(q;q)_n}\, {}_4\phi_3\left[\begin{matrix}q^{-n}, \beta^2 q^n, \beta^{1/2}e^{i\theta}, \beta^{1/2}e^{-i\theta}\\ \beta q^{1/2}, -\beta q^{1/2}, -\beta\end{matrix}; q, q\right] \tag{5.6}$$

is derived in Gasper [22], and a nonterminating q-analogue of Clausen's formula is derived in Gasper and Rahman [26].

6. q-Extensions of the Fields and Wimp expansion formula (1.9)

Verma [32] showed that the Fields and Wimp expansion formula (1.9) is a special case of the expansion

$$\begin{aligned}\sum_{j=0}^{n} A_j B_j \frac{(xw)^j}{j!} = \sum_{n=0}^{\infty} \frac{(-x)^n}{n!(c+n)_n} \sum_{k=0}^{\infty} \frac{A_{n+k}x^k}{k!(c+2n+1)_k}\\ \cdot \sum_{j=0}^{n} \frac{(-n)_j(c+n)_j}{j!}\, B_j w^j\end{aligned} \tag{6.1}$$

and derived the q-analogue

$$\begin{aligned}\sum_{j=0}^{\infty} A_j B_j \frac{(xw)^j}{(q;q)_j} = \sum_{n=0}^{\infty} \frac{(-x)^n q^{n(n-1)/2}}{(q;q)_n(cq^n;q)_n}\\ \cdot \sum_{k=0}^{\infty} \frac{A_{n+k}x^k}{(q;q)_k(cq^{2n+1};q)_k} \sum_{j=0}^{n} \frac{(q^{-n};q)_j(cq^n;q)_j}{(q;q)_j}\, B_j (wq)^j.\end{aligned} \tag{6.2}$$

From (6.2) it follows that (1.9) has a q-analogue of the form

$$(6.3)\qquad {}_{r+t}\phi_{s+u}\left[\begin{matrix} a_R, c_T \\ b_S, d_U \end{matrix}; q, xw\right]$$
$$= \sum_{n=0}^{\infty} \frac{(c_T; q)_n (e_K; q)_n x^n [(-1)^n q^{n(n-1)/2}]^{2+m+u-k-t}}{(q;q)_n (d_U; q)_n (f_M; q)_n (\gamma q^n; q)_n}$$
$$\cdot\ {}_{k+t}\phi_{m+u+1}\left[\begin{matrix} c_T q^n, e_K q^n \\ \gamma q^{2n+1}, d_U q^n, f_M q^n \end{matrix}; q, xq^{n(2+m+u-k-t)}\right]$$
$$\cdot\ {}_{m+r+2}\phi_{k+s}\left[\begin{matrix} q^{-n}, \gamma q^n, a_R, f_M \\ b_S, e_K \end{matrix}; q, wq\right]$$

where we used a contracted notation analogous to that used in (1.9).

To derive (6.2) first observe that since the identities (4.1) and (4.3) have the q-analogues

$$(6.4)\qquad (a;q)_{n+k} = (a;q)_n (aq^n; q)_k$$
$$(6.5)\qquad (q^{-n}; q)_k = (-1)^k q^{k(k-1)/2-nk} (q;q)_n / (q;q)_{n-k}$$

and since

$$(6.6)\qquad (a;q)_n = (1-a)(aq;q)_{n-1}$$

we can enter the definitions

In[1] := **asnpk[a_,q_,n_,k_] := o[a,q,n] o[a q^n,q,k]/o[a,q,n+k]**

In[2] := **mnsk[mn_,q_,k_] := (-1)^k q^(k(k-1)/2 +mn k) o[q,q,-mn]/**
(o[q,q,-mn-k] o[q^mn,q,k])

In[3] := **shift[a_,q_,n_] := (1-a) o[a q,q,n-1]/o[a,q,n]**

It should be noted that both of the functions asnpk[a,q,n,k] and asnpk[a,n,k] can be used in the same session because Mathematica will distinguish them by their different number of arguments. Similarly, both of the functions mnsk[mn,q,k] and mnsk[mn,k] can be used in the same Mathematica session.

So that we can also employ the $b \to 0$ limit case of the $p = q$ case of the summation formula (3.5), let's enter its k^{th} term in the form

In[4] := **deltasum[n_,a_,q_,k_] := (1-a q^(2k)) o[a,q,k] o[q^(-n),q,k] q^(n k)/**
((1-a) o[q,q,k] o[a q^(n+1),q,k])

Then the product $A[j]x^j$ can be represented by entering

In[5] := **deltasum[m,c q^(2j),q,i] A[j+m] x^(j+m)/(o[q,q,m] o[c q^(2j+1),q,m])**

whose sum over all $m \geq 0$ equals $A[j]x^j$, to get

$$\textit{Out[5]} = \frac{q^{i\ m}\ x^{j\ +\ m}\ A[j+m]\ (1 - c\ q^{2\ i\ +\ 2\ j})\ o[c\ q^{2\ j},\ q,\ i]\ o[q^{-m},\ q,\ i]}{(1 - c\ q^{2\ j})\ o[q,\ q,\ i]\ o[q,\ q,\ m]\ o[c\ q^{1\ +\ 2\ j},\ q,\ m]\ o[c\ q^{1\ +\ 2\ j\ +\ m},\ q,\ i]}$$

where i and m are the indexes of summation. Next use

In[6] := **% asnpk[c q^(2j+1),q,m,i] shift[c q^(2j),q,i] mnsk[-m,q,i]/ asnpk[c q^(2j+1),q,i-1,m+1]**

to obtain

$$Out[6] = \frac{(-1)^i \; q^{\frac{i\;(-1\;+\;i)}{2}} \; x^{j\;+\;m} \; A[j+m] \; (1 - c\; q^{2\;i\;+\;2\;j})}{o[q,\; q,\; i] \; o[q,\; q,\; -i+m] \; o[c\; q^{i\;+\;2\;j},\; q,\; 1+m]}$$

Applying the change in indexes of summation

In[7] := **Simplify**[**%/. {i->n-j, m->n+k-j}**]

we have

$$Out[7] = \frac{(-1)^{-j\;+\;n} \; q^{\frac{(-j\;+\;n)\;(-1\;-\;j\;+\;n)}{2}} \; x^{k\;+\;n} \; A[k+n] \; (1 - c\; q^{2\;n})}{o[q,\; q,\; k] \; o[q,\; q,\; -j+n] \; o[c\; q^{j\;+\;n},\; q,\; 1-j+k+n]}$$

where n and k are the indexes of summation. Then

In[8] := **Simplify**[**% asnpk[c q^n,q,j,n+k+1-j]/ (mnsk[-n,q,j] asnpk[c q^n,q,n,k+1] shift[c q^(2n),q,k+1])**]

gives

$$Out[8] = \frac{(-1)^{-2\;j\;+\;n} \; q^{j\;-\;\frac{n}{2}\;+\;\frac{n^2}{2}} \; x^{k\;+\;n} \; A[k+n] \; o[q^{-n},\; q,\; j] \; o[c\; q^{n},\; q,\; j]}{o[q,\; q,\; k] \; o[q,\; q,\; n] \; o[c\; q^{n},\; q,\; n] \; o[c\; q^{1\;+\;2\;n},\; q,\; k]}$$

Hence if we multiply by B[j] w^j /o[q,q, j] and sum from $j = 0$ to ∞ by using

In[9] := **% (-1)^(2j) B[j] w^j/o[q,q, j]**

we finally obtain

$$Out[9] = \frac{(-1)^n \; q^{j\;-\;\frac{n}{2}\;+\;\frac{n^2}{2}} \; w^j \; x^{k\;+\;n} \; A[k+n] \; B[j] \; o[q^{-n},\; q,\; j] \; o[c\; q^{n},\; q,\; j]}{o[q,\; q,\; j] \; o[q,\; q,\; k] \; o[q,\; q,\; n] \; o[c\; q^{n},\; q,\; n] \; o[c\; q^{1\;+\;2\;n},\; q,\; k]}$$

which gives the n, j, k^{th} term of the triple sum on the right side of (6.2) and so concludes our derivation of (6.2). This technique can also be employed to derive the extensions of (6.2) and (6.3) in [21,§4] and the bibasic extension in [21,§3] of Euler's transformation formula

$$\sum_{n=0}^{\infty} a_n b_n x^n = \sum_{k=0}^{\infty} (-1)^k \frac{x^k}{k!} f^{(k)}(x) \Delta^k a_0 \tag{6.7}$$

where

$$f(x) = b_0 + b_1 x + b_2 x^2 + \cdots$$

and

$$\Delta^k a_0 = \sum_{j=0}^{k} (-1)^j \binom{k}{j} a_{k-j}.$$

The Fields and Wimp expansion formula (1.9) follows from (6.3) by replacing each parameter in (6.3) by a power of q and letting $q \to 1$. In [18,23] it was pointed out that by using (5.4) in (1.9) we obtain the sum of squares expansion

$$
\begin{aligned}
(6.8)\qquad & {}_3F_2\left[\begin{matrix}-n, n+a+2, (a+1)/2 \\ a+1, (a+3)/2\end{matrix}; (1-x^2)(1-y^2)\right] \\
&= \sum_{k=0}^{n} \frac{n!(n+a+2)_k((a+2)/2)_k}{k!(n-k)!((a+3)/2)_k(k+a+1)_k}\,(1-y^2)^k \\
&\quad \cdot \left\{ \frac{k!(n-k)!}{(a+1)_k(2k+a+2)_{n-k}}\, C_k^{(a+1)/2}(x) C_{n-k}^{k+(a+2)/2}(y) \right\}^2
\end{aligned}
$$

which immediately gives the Askey-Gasper inequality (1.10) since each term on the right side of (6.8) is clearly nonnegative. The special case $y = 0$ of (6.8) gives the expansion in [3, (1.16)]. For a q-extension of (6.8), see [22].

7. Jensen's necessary and sufficient conditions for the Riemann Hypothesis

Among Jensen's necessary and sufficient conditions for the Riemann Hypothesis given in Pólya [28] is the condition that

$$
\int_{-\infty}^{\infty}\int_{-\infty}^{\infty} \Phi(s)\Phi(t)e^{i(s+t)x}(s-t)^{2n}\,ds\,dt \ge 0 \tag{7.1}
$$

for all real x when $n = 0, 1, 2, \ldots$, where

$$
\Phi(t) = 4\sum_{k=1}^{\infty}(2k^4\pi^2e^{9t} - 3k^2\pi e^{5t})e^{-k^2\pi e^{4t}} \tag{7.2}
$$

is an even function of t which is positive for all real t. Fourteen years ago, I pointed out in a survey paper [18, §9] on positivity and special functions that, since the above integral is a square when $n = 0$, the method of sums of squares (discussed earlier in the paper) is suggested for proving (7.1). I also stated that a computer analysis of (7.1) and of the other necessary and sufficient conditions for the Riemann Hypothesis in [28] might lead to some interesting observations.

Recently, Csordas and Varga [12] (also see [13]) considered the inequalities [28, (18)]

$$
\int_{-\infty}^{\infty}\int_{-\infty}^{\infty} \Phi(s)\Phi(t)e^{i(s+t)x}e^{(s-t)y}(s-t)^2\,ds\,dt \ge 0 \tag{7.3}
$$

for real x and y, which is one of Jensen's necessary and sufficient conditions for the Riemann Hypotheses to hold, and showed that it suffices to prove (7.3) for $0 \le x < \infty$ when y is in the bounded interval $0 \le y < 1$. In view of the maximum principles that are known to hold for certain kernels involving orthogonal polynomials and other special functions [18, §5], this suggests the conjecture that if the inequalities in (7.3) hold for $0 \le x < \infty$ when $y = 0$ then they hold for all real x and y. A proof of this conjecture would reduce proving

(7.3) for $0 \le x < \infty$, $0 \le y < 1$ to just proving the more tractable single variable special case $0 \le x < \infty$, $y = 0$.

When $n = 1$ in (7.1) or, equivalently, $y = 0$ in (7.3), the evenness of $\Phi(t)$ and the identity $e^{i\theta} = \cos\theta + i\sin\theta$ can be used to show that

$$(7.4)\quad \frac{1}{8}\int_{-\infty}^{\infty}\int_{-\infty}^{\infty}\Phi(s)\Phi(t)e^{i(s+t)x}(s-t)^2 ds dt$$
$$= \left(\int_0^{\infty}\Phi(t)\cos(xt)dt\right)\left(\int_0^{\infty}\Phi(t)t^2\cos(xt)dt\right) + \left(\int_0^{\infty}\Phi(t)t\sin(xt)dt\right)^2.$$

An extension of (7.4) to arbitrary y is derived in [12]. The main advantage of (7.4) is that it reduces the computation of the double integral to that of the single integrals on the right side, which takes a lot less time when using numerical integration to approximate the integrals for particular values of x. Since, as expected, the use of Mathematica's numerical integration NIntegrate function (with appropriate truncations of the series (7.2) and range of integration and suitable settings of the WorkingPrecision, AccuracyGoal, MinRecursion, MaxRecursion, and Points options) only gave positive values for the right side of (7.4) for each chosen value of x, I decided to investigate what happens when the function $\Phi(t)$ in the integrals on the right side of (7.4) is replaced by the n^{th} partial sum (call it $\Phi_n(t)$) of the series representation for $\Phi(t)$ in (7.2). Letting $f_n(x)$ denote the right side of (7.4) with $\Phi(t)$ replaced by $\Phi_n(t)$, it was found that $f_1(x)$ changes sign from positive to negative in the interval (37, 38), and that when $n = 2, 3, 4, 5, 6, 7, 8,$ and 9 the functions $f_n(x)$ change sign from positive to negative in the intervals (85, 86), (134, 135), (210, 211), (302, 303), (401, 402), (519, 520), (657, 658), and (817, 818), respectively. I conjecture that for each natural number n there is an $x_n > 0$ such that $x_n \to \infty$ and $f_n(x) \ge 0$ for $0 \le x \le x_n$, which would imply that (7.1) and (7.3) hold for real x when $n = 1$ and $y = 0$.

REFERENCES

[1] R. Askey, 'Orthogonal polynomials and positivity,' *Studies in Applied Mathematics* **6**, Special Functions and Wave Propagation (D. Ludwig and F. W. J. Olver, eds.), SIAM, Philadelphia, 1970, pp. 64–85.

[2] R. Askey and G. Gasper, 'Jacobi polynomial expansions of Jacobi polynomials with non-negative coefficients,' *Proc. Camb. Phil. Soc.* **70** (1971), 243–255.

[3] R. Askey and G. Gasper, 'Positive Jacobi polynomial sums II,' *Amer. J. Math.* **98** (1976), 709–737.

[4] R. Askey and G. Gasper, 'Inequalities for polynomials,' *The Bieberbach Conjecture: Proc. of the Symposium on the Occasion of the Proof* (A. Baernstein, D. Drasin, P. Duren, and A. Marden, eds.), Math. Surveys and Monographs **21**, Amer. Math. Soc., Providence, R. I., 1986, pp. 7–32.

[5] R. Askey and J. Wilson, 'Some basic hypergeometric polynomials that generalize Jacobi polynomials,' *Memoirs Amer. Math. Soc.* **319**, Amer. Math. Soc., Providence, R. I., 1985.

[6] W. N. Bailey, *Generalized Hypergeometric Series*, Cambridge University Press, Cambridge, 1935; reprinted by Stechert-Hafner, New York, 1964.

[7] L. de Branges, 'A proof of the Bieberbach conjecture,' *Acta Math.* **154** (1985), 137–152.

[8] L. de Branges, 'Powers of Riemann mapping functions,' *The Bieberbach Conjecture: Proc. of the Symposium on the Occasion of the Proof* (A. Baernstein, et al., eds.), Math. Surveys and Monographs **21**, Amer. Math. Soc., Providence, R. I., 1986, pp. 51–67.

[9] D. M. Bressoud, 'The Bailey Lattice: an introduction,' *Ramanujan Revisited* (G. E. Andrews et al., eds.), Academic Press, New York, 1988, pp. 57–67.

[10] T. S. Chihara, *An Introduction to Orthogonal Polynomials*, Gordon and Breach, New York, 1978.

[11] T. Clausen, 'Ueber die Fälle, wenn die Reihe von der Form ... ein Quadrat von der Form ... hat,' *J. reine angew. Math.* **3** (1828), 89–91.

[12] G. Csordas and R. S. Varga, 'Fourier transforms and the Hermite-Biehler theorem,' to appear.

[13] G. Csordas and R. S. Varga, 'Necessary and sufficient conditions and the Riemann Hypothesis,' to appear.

[14] A. Erdélyi, *Higher Transcendental Functions, Vols. I & II*, McGraw-Hill, New York, 1953.

[15] J. L. Fields and J. Wimp, 'Expansions of hypergeometric functions in hypergeometric functions,' *Math. Comp.* **15** (1961), 390–395.

[16] R. Gangolli, 'Positive definite kernels on homogeneous spaces and certain stochastic processes related to Lévy's Brownian motion of several parameters,' *Ann. Inst. H. Poincaré, Sect. B Vol. III*, 1967, pp. 121–226.

[17] G. Gasper, 'Projection formulas for orthogonal polynomials of a discrete variable,' *J. Math. Anal. Appl.* **45** (1974), 176–198.

[18] G. Gasper, 'Positivity and special functions,' *Theory and Applications of Special Functions* (R. Askey, ed.), Academic Press, New York, 1975, pp. 375–433.

[19] G. Gasper, 'Positive sums of the classical orthogonal polynomials,' *SIAM J. Math. Anal.* **8** (1977), 423–447.

[20] G. Gasper, 'A short proof of an inequality used by de Branges in his proof of the Bieberbach, Robertson and Milin conjectures,' *Complex Variables* **7** (1986), 45–50.

[21] G. Gasper, 'Summation, transformation, and expansion formulas for bibasic series,' *Trans. Amer. Math. Soc.* **312** (1989), 257–277.

[22] G. Gasper, 'q-Extensions of Clausen's formula and of the inequalities used by de Branges in his proof of the Bieberbach, Robertson, and Millin conjectures,' *SIAM J. Math. Anal.* **20** (1989), 1019–1034.

[23] G. Gasper, 'Bibasic summation, transformation and expansion formulas, q-analogues of Clausen's formula, and nonnegative basic hypergeometric series,' *Workshop on q-Series and Partitions* (D. Stanton, ed.), IMA Volumes in Mathematics and Its Applications, Springer, Berlin and New York, to appear.

[24] G. Gasper and M. Rahman, 'Product formulas of Watson, Bailey and Bateman types and positivity of the Poisson kernel for q-Racah polynomials,' *SIAM J. Math. Anal.* **15** (1984), 768–789.

[25] G. Gasper and M. Rahman, 'An indefinite bibasic summation formula and some quadratic, cubic, and quartic summation and transformation formulas,' *Canad. J. Math.*, to appear.

[26] G. Gasper and M. Rahman, 'A nonterminating q-Clausen formula and some related product formulas,' *SIAM J. Math. Anal.*, to appear.

[27] G. Gasper and M. Rahman, *Basic Hypergeometric Series*, Cambridge University Press, Cambridge, to appear.

[28] G. Pólya, 'Über die algebraisch-funktionentheoretischen Untersuchungen von J. L. W. V. Jensen,' *Kgl. Danske Videnskabernes Selskab. Math.-Fys. Medd.* **7** (17) (1927), pp. 3–33; reprinted in his Collected Papers, Vol. II, pp. 278–308.

[29] D. B. Sears, 'On the transformation theory of basic hypergeometric functions,' *Proc. London Math. Soc.* (2) **53** (1951), 158–180.

[30] L. J. Slater, *Generalized Hypergeometric Functions*, Cambridge University Press, Cambridge, 1966.

[31] G. Szegö, *Orthogonal Polynomials*, 4th edition, Amer. Math. Soc. Colloq. Publ. 23, 1975.

[32] A. Verma, 'Certain expansions of the basic hypergeometric functions,' *Math. Comp.* **20** (1966), 151–157.